

Module 04 — Formules essentielles

Bootcamp Data Analyst — From Zero to Hero | Niveau Débutant · Acte II

Ce que tu seras capable de faire à la fin de ce module

- Utiliser les 11 formules incontournables d'un DA au quotidien
- Combiner des formules pour répondre à des questions business concrètes
- Gérer les erreurs dans les formules avec `SIERREUR()` / `IFERROR()`
- Consulter la cheatsheet pour aller plus loin

 **Durée estimée** : 60 minutes

 **Prérequis** : Module 03 complété — références absolues/relatives maîtrisées.

 **Convention** : toutes les formules sont données en **FR / EN** — Excel adapte selon la langue de ton interface, Google Sheets accepte les deux.

 **Séparateur d'arguments** : en français (et dans la majorité des pays africains), le séparateur entre arguments est le **point-virgule** `;`. En anglais, c'est la **virgule** `,`. Si tu copies une formule depuis Stack Overflow ou une doc anglaise et qu'elle ne fonctionne pas, c'est souvent ça. Exemples : `=SOMME.SI(B2:B10 ; "Abidjan" ; D2:D10)` en FR · `=SUMIF(B2:B10, "Abidjan", D2:D10)` en EN. Dans ce module, on utilise le `;`.

1. Le dataset fil rouge

Pour tout ce module, on travaille avec le même dataset : **les ventes mensuelles d'un opérateur télécom fictif**, 3 régions, 3 produits.

	A	B	C	D	E
1	Vendeur	Région	Produit	Ventes	Objectif
2	Adjoua	Abidjan	Recharge 500	1 200 000	1 000 000
3	Brice	Bouaké	Forfait inter	980 000	1 200 000
4	Camille	Abidjan	Forfait voix	1 500 000	1 000 000
5	Drissa	San Pedro	Recharge 500	620 000	800 000
6	Estelle	Bouaké	Forfait inter	1 100 000	1 200 000
7	Fanta	Abidjan	Recharge 500	890 000	1 000 000
8	Gilles	San Pedro	Forfait voix	750 000	800 000
9	Hervé	Abidjan	Forfait inter	1 350 000	1 000 000
10	Inès	Bouaké	Recharge 500	540 000	800 000

💡 Crée ce tableau dans Excel ou Google Sheets et applique chaque formule en temps réel.
C'est 10x plus efficace que de lire passivement.

2. **SOMME()** / **SUM()** — additionner

Syntaxe

```
=SOMME(plage)      =SUM(range)  
=SOMME(D2:D10)     =SUM(D2:D10)
```

Exemples sur notre dataset

```
=SOMME(D2:D10)  → Total des ventes de toute l'équipe  
=SOMME(D2,D4,D9) → Ventes des vendeurs Abidjan uniquement (sélection manuelle)
```

Pièges

- `SOMME()` / `SUM()` ignore les cellules contenant du **texte** — si le total semble trop bas, vérifier les types
- `SOMME()` / `SUM()` inclut les cellules **masquées** par un filtre — utiliser `SOUS.TOTAL()` / `SUBTOTAL()` pour sommer uniquement les lignes visibles

💡 *Raccourci : sélectionner une colonne de nombres et regarder la barre d'état en bas — Excel/Sheets affiche la somme automatiquement sans formule.*

3. `MOYENNE()` / `AVERAGE()` — `MIN()` — `MAX()`

Syntaxe

```
=MOYENNE(D2:D10)    =AVERAGE(D2:D10)    → Ventes moyennes
=MIN(D2:D10)         → Vente la plus basse
=MAX(D2:D10)         → Vente la plus haute
```

Exemples sur notre dataset

```
=MOYENNE(D2:D10)    → 958 889 FCFA (vente moyenne par vendeur)
=MIN(D2:D10)        → 540 000 FCFA (Inès, Bouaké)
=MAX(D2:D10)        → 1 500 000 FCFA (Camille, Abidjan)
```

Rappel du Module 02

La moyenne est sensible aux outliers. Si un vendeur a des ventes exceptionnelles (ou nulles), elle sera faussée. Dans ce cas, préférer la médiane avec `MEDIANE()` / `MEDIAN()`.

4. `NB.SI()` / `COUNTIF()` — compter selon une condition

Syntaxe

```
=NB.SI(plage_critère ; critère)  
=COUNTIF(range ; criteria)
```

Exemples sur notre dataset

```
=NB.SI(B2:B10 ; "Abidjan")  
=COUNTIF(B2:B10 ; "Abidjan")  
→ Combien de vendeurs sont basés à Abidjan ? → 4  
  
=NB.SI(D2:D10 ; ">1000000")  
=COUNTIF(D2:D10 ; ">1000000")  
→ Combien de vendeurs dépassent 1 million de ventes ? → 4  
  
=NB.SI(C2:C10 ; "Recharge 500")  
=COUNTIF(C2:C10 ; "Recharge 500")  
→ Combien de vendeurs vendent la Recharge 500 ? → 3
```

Les opérateurs de comparaison disponibles

">1000000"	supérieur à
">=1000000"	supérieur ou égal
"<1000000"	inférieur à
"<>Abidjan"	différent de
"Abi*"	commence par "Abi" (joker *)

💡 Pour compter avec **plusieurs conditions** → `NB.SI.ENS()` / `COUNTIFS()` (dans la cheatsheet)

5. `SOMME.SI()` / `SUMIF()` — additionner selon une condition

Syntaxe

```
=SOMME.SI(plage_critère ; critère ; plage_somme)  
=SUMIF(range ; criteria ; sum_range)
```

Exemples sur notre dataset

```
=SOMME.SI(B2:B10 ; "Abidjan" ; D2:D10)  
=SUMIF(B2:B10 ; "Abidjan" ; D2:D10)  
→ Total des ventes uniquement pour la région Abidjan  
→ = 1 200 000 + 1 500 000 + 890 000 + 1 350 000 = 4 940 000 FCFA  
  
=SOMME.SI(D2:D10 ; ">1000000" ; D2:D10)  
=SUMIF(D2:D10 ; ">1000000" ; D2:D10)  
→ Somme des ventes supérieures à 1 million seulement
```

Cas d'usage business

- Chiffre d'affaires par région
- Total des ventes d'un produit spécifique
- Somme des transactions au-dessus d'un seuil

💡 Pour sommer avec **plusieurs conditions** → `SOMME.SI.ENS()` / `SUMIFS()` (dans la cheatsheet)

6. `NBVAL()` / `COUNTA()` et `NB.VIDE()` / `COUNTBLANK()` — contrôle qualité

Avant toute analyse, la première question d'un DA c'est : **est-ce que mes données sont complètes ?** Ces deux formules sont le premier réflexe.

Syntaxe

<code>=NBVAL(plage)</code>	<code>=COUNTA(range)</code>	→ Compte les cellules NON vides
<code>=NB.VIDE(plage)</code>	<code>=COUNTBLANK(range)</code>	→ Compte les cellules VIDES
<code>=NB(plage)</code>	<code>=COUNT(range)</code>	→ Compte les cellules contenant un NOMBRE

Exemples sur notre dataset

<code>=NBVAL(A2:A10)</code>	<code>=COUNTA(A2:A10)</code>
→ 9 – tous les vendeurs ont un nom renseigné ✓	
<code>=NB.VIDE(D2:D100)</code>	<code>=COUNTBLANK(D2:D100)</code>
→ 0 – aucune vente manquante ✓	
→ 12 – 12 lignes sans valeur ⚠ à investiguer	
<code>=NB(D2:D10)</code>	<code>=COUNT(D2:D10)</code>
→ 9 – toutes les cellules Ventes sont bien des nombres	
→ 7 – 2 cellules contiennent du texte au lieu de nombres ⚠	

Tableau de contrôle qualité — à faire sur chaque nouveau dataset

Colonne	Total lignes	Remplies	Vides
Vendeur	<code>=NBVAL(A:A)-1</code>	<code>=NBVAL(A2:A1000)</code>	
	<code>=NB.VIDE(A2:A1000)</code>		
Région	<code>=NBVAL(A:A)-1</code>	<code>=NBVAL(B2:B1000)</code>	
	<code>=NB.VIDE(B2:B1000)</code>		
Ventes	<code>=NBVAL(A:A)-1</code>	<code>=NB(D2:D1000)</code>	
	<code>=NB.VIDE(D2:D1000)</code>		

💡 Ce petit tableau de contrôle qualité en haut de chaque feuille d'analyse te permet de détecter immédiatement les colonnes incomplètes — avant de calculer des moyennes ou des totaux sur des données trouées.

7. MOYENNE.SI() / AVERAGEIF() — moyenne selon une condition

Syntaxe

```
=MOYENNE.SI(plage_critère ; critère ; plage_moyenne)  
=AVERAGEIF(range ; criteria ; average_range)
```

Exemples sur notre dataset

```
=MOYENNE.SI(B2:B10 ; "Abidjan" ; D2:D10)  
=AVERAGEIF(B2:B10 ; "Abidjan" ; D2:D10)  
→ Ventes moyennes des vendeurs basés à Abidjan  
→ = (1 200 000 + 1 500 000 + 890 000 + 1 350 000) / 4 = 1 235 000 FCFA  
  
=MOYENNE.SI(C2:C10 ; "Forfait inter" ; D2:D10)  
=AVERAGEIF(C2:C10 ; "Forfait inter" ; D2:D10)  
→ Ventes moyennes sur le produit Forfait internet
```

Pourquoi c'est puissant

Ces trois formules — **NB.SI**, **SOMME.SI**, **MOYENNE.SI** — te permettent de faire des **agrégations par groupe** directement dans Excel, sans TCD. C'est l'équivalent d'un **GROUP BY** en SQL que tu verras en Acte III.

8. **SI()** / **IF()** — logique conditionnelle

Syntaxe

```
=SI(condition ; valeur_si_vrai ; valeur_si_faux)  
=IF(condition ; value_if_true ; value_if_false)
```

Exemples sur notre dataset

```
=SI(D2>=E2 ; "✅ Objectif atteint" ; "❌ En dessous")  
=IF(D2>=E2 ; "✅ Objectif atteint" ; "❌ En dessous")  
→ Adjoua : 1 200 000 >= 1 000 000 → "✅ Objectif atteint"  
→ Brice : 980 000 < 1 200 000 → "❌ En dessous"  
  
=SI(D2>=1200000 ; "Top" ; SI(D2>=900000 ; "Moyen" ; "Faible"))  
=IF(D2>=1200000 ; "Top" ; IF(D2>=900000 ; "Moyen" ; "Faible"))  
→ SI imbriqué – 3 niveaux de performance
```

Combiner avec ET() / AND() et OU() / OR()

```
=SI(ET(D2>1000000 ; B2="Abidjan") ; "Bon vendeur Abidjan" ; "")  
=IF(AND(D2>1000000 ; B2="Abidjan") ; "Bon vendeur Abidjan" ; "")  
→ Condition vraie uniquement si LES DEUX conditions sont vraies  
  
=SI(OU(B2="Abidjan" ; B2="Bouaké") ; "Zone Nord" ; "Zone Sud")  
=IF(OR(B2="Abidjan" ; B2="Bouaké") ; "Zone Nord" ; "Zone Sud")  
→ Condition vraie si AU MOINS UNE condition est vraie
```

⚠️ Ne pas imbriquer plus de 3-4 SI — ça devient illisible. Pour des conditions complexes, utiliser **SI.CONDITIONS()** / **IFS()** ou une colonne de référence.

9. `SIERREUR()` / `IFERROR()` — gérer les erreurs

Syntaxe

```
=SIERREUR(formule ; valeur_si_erreur)  
=IFERROR(formula ; value_if_error)
```

Pourquoi c'est indispensable

Sans `SIERREUR` / `IFERROR`, une seule erreur `#N/A` ou `#DIV/0!` dans une colonne peut **casser tout un tableau de bord**. Cette formule remplace l'erreur par une valeur de ton choix.

```
=SIERREUR(D2/E2 ; 0)  
=IFERROR(D2/E2 ; 0)  
→ Si E2 = 0, retourne 0 au lieu de #DIV/0!
```



```
=SIERREUR(RECHERCHEV(A2 ; catalogue ; 2 ; 0) ; "Produit introuvable")  
=IFERROR(VLOOKUP(A2 ; catalogue ; 2 ; 0) ; "Produit introuvable")  
→ Si la RECHERCHEV ne trouve pas la valeur, affiche un message clair au lieu de #N/A
```

💡 Règle professionnelle : toute formule `RECHERCHEV` / `VLOOKUP` doit être enveloppée dans un `SIERREUR` / `IFERROR`. Sans ça, ton fichier sera plein de `#N/A` à la première valeur manquante.

10. `RECHERCHEV()` / `VLOOKUP()` — chercher une valeur dans un tableau

C'est la formule la plus puissante et la plus utilisée en analyse de données. Elle permet de **rapprocher deux tableaux** en cherchant une valeur commune — l'équivalent d'un **JOIN** en SQL.

Syntaxe

```
=RECHERCHEV(valeur_cherchée ; tableau ; numéro_colonne ;  
correspondance_exacte)  
=VLOOKUP(lookup_value ; table_array ; col_index ; range_lookup)
```

ARGUMENT	RÔLE	VALEUR
<code>valeur_cherchée</code>	Ce qu'on cherche	Une cellule ou une valeur
<code>tableau</code>	Où chercher	Une plage ou plage nommée
<code>numéro_colonne</code>	Quelle colonne retourner	1 = première colonne du tableau
<code>correspondance_exacte</code>	Type de correspondance	0 ou FAUX = exacte (presque toujours)

Exemple concret

Tu as deux tableaux : les ventes (col A = Vendeur) et un tableau RH (col A = Vendeur, col B = Département).

Tableau ventes (A:E)		Tableau RH (G:H)	
A	B	G	H
Vendeur	Région	Vendeur	Département
Adjoua	Abidjan	Adjoua	Commercial
Brice	Bouaké	Brice	Technique
...		...	

→ En F2 : =RECHERCHEV(A2 ; \$G\$2:\$H\$10 ; 2 ; 0)
=VLOOKUP(A2 ; \$G\$2:\$H\$10 ; 2 ; 0)
→ Retourne "Commercial" pour Adjoua

Règles et limites

- La valeur cherchée **doit être dans la première colonne** du tableau de recherche
- **RECHERCHEV** / **VLOOKUP** ne cherche **que vers la droite** — pour chercher à gauche, utiliser **INDEX/EQUIV** / **INDEX/MATCH**
- Toujours utiliser **0** ou **FAUX** comme dernier argument pour une correspondance exacte
- Toujours figer le tableau avec **\$** : **\$G\$2:\$H\$10** pour pouvoir copier la formule
- Envelopper dans **SIERREUR** / **IFERROR** pour gérer les valeurs introuvables

Version professionnelle complète :

```
=SIERREUR(RECHERCHEV(A2 ; $G$2:$H$10 ; 2 ; 0) ; "Non trouvé")
```

```
=IFERROR(VLOOKUP(A2 ; $G$2:$H$10 ; 2 ; 0) ; "Not found")
```



RECHERCHEV / **VLOOKUP** est l'équivalent Excel d'un **LEFT JOIN** en SQL. Quand tu apprendras SQL en Acte III, tu reconnaîtras exactement la même logique.

11. **CONCATENER()** / **CONCAT()** / **&** — assembler du texte

Syntaxe — 3 façons de faire la même chose

```
=CONCATENER(A2 ; " - " ; B2)
```

ancienne syntaxe, toujours valide

```
=CONCAT(A2 ; " - " ; B2)
```

nouvelle syntaxe (Excel 2019+, Sheets)

```
=A2 & " - " & B2
```

opérateur & — le plus rapide à écrire

Exemples sur notre dataset

```
=A2 & " (" & B2 & ")"
```

```
→ "Adjoua (Abidjan)"
```

```
=B2 & " - " & C2
```

```
→ "Abidjan - Recharge 500"
```

```
= "Ventes " & B2 & " : " & TEXTE(D2 ; "# ##0") & " FCFA"
```

```
= " Sales " & B2 & " : " & TEXT(D2 ; "#,##0") & " CFA"
```

```
→ "Ventes Abidjan : 1 200 000 FCFA"
```

Cas d'usage

- Créer une clé unique pour rapprocher deux tableaux : `=A2 & "_" & B2` → `"Adjoua_Abidjan"`
- Générer des messages dynamiques dans un dashboard
- Formater des étiquettes pour des graphiques

💡 L'opérateur `&` est le plus utilisé en pratique — plus rapide à écrire que `CONCATENER` / `CONCAT`.

12. `SUPPRESPACE()` / `TRIM()` et `AUJOURDHUI()` / `TODAY()`

`SUPPRESPACE()` / `TRIM()` — nettoyer les espaces

Supprime les espaces en début et fin de cellule, et les espaces doubles internes.

`=SUPPSPACE(A2)` `=TRIM(A2)`

" Abidjan "	→	"Abidjan"
"Bouaké "	→	"Bouaké"
"San Pedro"	→	"San Pedro"

C'est l'une des premières formules à appliquer quand tu reçois un fichier externe — les espaces invisibles cassent les **RECHERCHEV** et les filtres.

AUJOURDHUI() / **TODAY()** — date du jour dynamique

Retourne la date du jour, mise à jour automatiquement à chaque ouverture du fichier.

<code>=AUJOURDHUI()</code>	<code>=TODAY()</code>
→ 2025-03-15	(date du jour)
<code>=AUJOURDHUI() - D2</code>	<code>=TODAY() - D2</code>
→ Nombre de jours depuis la date en D2	
<code>=ANNEE(AUJOURDHUI())</code>	<code>=YEAR(TODAY())</code>
→ 2025	
<code>=MOIS(AUJOURDHUI())</code>	<code>=MONTH(TODAY())</code>
→ 3 (mars)	

Cas d'usage

- Calculer l'ancienneté d'un employé : `=AUJOURDHUI() - date_embauche`
- Signaler les contrats qui expirent dans moins de 30 jours
- Créer des dashboards avec une date de mise à jour automatique



MAINTENANT() / **NOW()** retourne la date ET l'heure. **AUJOURDHUI()** / **TODAY()** retourne seulement la date.

13. Combiner les formules — cas pratiques

En pratique, un DA combine rarement une seule formule. Voici des exemples réels.

Cas 1 — Taux de réalisation de l'objectif

```
=SIERREUR(D2/E2 ; 0)  
=IFERROR(D2/E2 ; 0)  
→ 1 200 000 / 1 000 000 = 1.2 → formater en % → 120%
```

Cas 2 — Statut performance avec seuils

```
=SI(D2/E2>=1.1 ; "🟢 Excellent" ; SI(D2/E2>=0.9 ; "🟡 OK" ; "🔴 Faible"))  
=IF(D2/E2>=1.1 ; "🟢 Excellent" ; IF(D2/E2>=0.9 ; "🟡 OK" ; "🔴 Faible"))
```

Cas 3 — Résumé par région

Région	Total ventes	Nb vendeurs
Moyenne		
Abidjan	=SOMME.SI(B:B;"Abidjan";D:D) =MOYENNE.SI(B:B;"Abidjan";D:D)	=NB.SI(B:B;"Abidjan")
Bouaké	=SOMME.SI(B:B;"Bouaké";D:D) =MOYENNE.SI(B:B;"Bouaké";D:D)	=NB.SI(B:B;"Bouaké")
San Pedro	=SOMME.SI(B:B;"San Pedro";D:D) =MOYENNE.SI(B:B;"San Pedro";D:D)	=NB.SI(B:B;"San Pedro")

Ce tableau de synthèse par région, c'est exactement ce qu'un manager attend d'un DA en début de semaine.

Cas 4 — Enrichissement avec RECHERCHEV + SIERREUR

```
=SIERREUR(RECHERCHEV(A2 ; tableau_rh ; 2 ; 0) ; "Non renseigné")  
=IFERROR(VLOOKUP(A2 ; hr_table ; 2 ; 0) ; "Not found")  
→ Ajouter le département de chaque vendeur depuis un tableau RH séparé
```

14. Cheatsheet — toutes les formules utiles

Les formules ci-dessous ne sont pas toutes dans le module mais tu en auras besoin sur le terrain. Garde cette section en signet.

Calcul & Statistiques

FRANÇAIS	ANGLAIS	RÔLE
SOMME()	SUM()	Somme d'une plage
MOYENNE()	AVERAGE()	Moyenne
MEDIANE()	MEDIAN()	Médiane
MIN()	MIN()	Valeur minimale
MAX()	MAX()	Valeur maximale
GRANDE.VALEUR()	LARGE()	Nème plus grande valeur
PETITE.VALEUR()	SMALL()	Nème plus petite valeur
ECARTYPE()	STDEV()	Écart-type
SOUS.TOTAL()	SUBTOTAL()	Calcul sur lignes visibles
ARRONDI()	ROUND()	Arrondir à N décimales

Comptage & Conditions

FRANÇAIS	ANGLAIS	RÔLE
<code>NB()</code>	<code>COUNT()</code>	Compter les cellules numériques
<code>NBVAL()</code>	<code>COUNTA()</code>	Compter les cellules non vides
<code>NB.VIDE()</code>	<code>COUNTBLANK()</code>	Compter les cellules vides
<code>NB.SI()</code>	<code>COUNTIF()</code>	Compter selon 1 condition
<code>NB.SI.ENS()</code>	<code>COUNTIFS()</code>	Compter selon N conditions
<code>SOMME.SI()</code>	<code>SUMIF()</code>	Sommer selon 1 condition
<code>SOMME.SI.ENS()</code>	<code>SUMIFS()</code>	Sommer selon N conditions
<code>MOYENNE.SI()</code>	<code>AVERAGEIF()</code>	Moyenne selon 1 condition
<code>MOYENNE.SI.ENS()</code>	<code>AVERAGEIFS()</code>	Moyenne selon N conditions

Logique

FRANÇAIS	ANGLAIS	RÔLE
<code>SI()</code>	<code>IF()</code>	Condition simple
<code>SI.CONDITIONS()</code>	<code>IFS()</code>	Conditions multiples (sans imbrication)
<code>ET()</code>	<code>AND()</code>	Toutes les conditions vraies
<code>OU()</code>	<code>OR()</code>	Au moins une condition vraie
<code>NON()</code>	<code>NOT()</code>	Inverse une condition
<code>SIERREUR()</code>	<code>IFERROR()</code>	Gérer les erreurs
<code>SIVIDE()</code>	<code>IFNA()</code>	Gérer les erreurs #N/A

Recherche & Référence

FRANÇAIS	ANGLAIS	RÔLE
RECHERCHEV()	VLOOKUP()	Chercher vers la droite
RECHERCHEH()	HLOOKUP()	Chercher vers le bas
INDEX()	INDEX()	Retourner une valeur par position
EQUIV()	MATCH()	Trouver la position d'une valeur
RECHERCHEX()	XLOOKUP()	Version moderne de RECHERCHEV (Excel 365)
DECALER()	OFFSET()	Référencer une plage décalée

Texte

FRANÇAIS	ANGLAIS	RÔLE
CONCATENER()	CONCAT()	Assembler du texte
JOINDRE , TEXTE()	TEXTJOIN()	Joindre avec séparateur
GAUCHE()	LEFT()	N caractères à gauche
DROITE()	RIGHT()	N caractères à droite
STXT()	MID()	Extraire du texte au milieu
NBCAR()	LEN()	Longueur d'un texte
MAJUSCULE()	UPPER()	Tout en majuscules
MINUSCULE()	LOWER()	Tout en minuscules
NOMPROPRE()	PROPER()	Première lettre en majuscule
SUPPRESSESPACE()	TRIM()	Supprimer espaces inutiles
SUBSTITUE()	SUBSTITUTE()	Remplacer du texte
TEXTE()	TEXT()	Formater un nombre en texte
VALEUR()	VALUE()	Convertir texte en nombre
CHERCHE()	SEARCH()	Trouver position d'un texte

Dates & Heures

FRANÇAIS	ANGLAIS	RÔLE
AUJOURDHUI()	TODAY()	Date du jour
MAINTENANT()	NOW()	Date et heure actuelles
ANNEE()	YEAR()	Extraire l'année
MOIS()	MONTH()	Extraire le mois
JOUR()	DAY()	Extraire le jour
JOURSEM()	WEEKDAY()	Jour de la semaine
NO.SEMAINES()	WEEKNUM()	Numéro de semaine
DATE()	DATE()	Créer une date depuis année/mois/jour
DATEDIF()	DATEDIF()	Différence entre deux dates
FIN.MOIS()	EOMONTH()	Dernier jour du mois

15. Résumé du module

FORMULE FR	FORMULE EN	RÔLE CLÉ
<code>SOMME()</code>	<code>SUM()</code>	Additionner une plage
<code>MOYENNE()</code> / <code>MIN()</code> / <code>MAX()</code>	<code>AVERAGE()</code> / <code>MIN()</code> / <code>MAX()</code>	Stats de base
<code>NB.SI()</code>	<code>COUNTIF()</code>	Compter selon condition
<code>SOMME.SI()</code>	<code>SUMIF()</code>	Sommer selon condition
<code>MOYENNE.SI()</code>	<code>AVERAGEIF()</code>	Moyenne selon condition
<code>SI()</code>	<code>IF()</code>	Logique conditionnelle
<code>SIERREUR()</code>	<code>IFERROR()</code>	Gérer les erreurs
<code>RECHERCHEV()</code>	<code>VLOOKUP()</code>	Rapprocher deux tableaux
<code>CONCATENER()</code> / <code>&</code>	<code>CONCAT()</code> / <code>&</code>	Assembler du texte
<code>SUPPRESSESPACE()</code>	<code>TRIM()</code>	Nettoyer les espaces
<code>AUJOURDHUI()</code>	<code>TODAY()</code>	Date dynamique

Quiz — Vérifie ta compréhension

Réponds mentalement, puis clique sur **"Voir la réponse"** pour te corriger.

Q1. Tu veux calculer le total des ventes uniquement pour la région Bouaké. Quelle formule utiliser ?

- a) `=SOMME(D2:D10)`
- b) `=SOMME.SI(B2:B10 ; "Bouaké" ; D2:D10)`
- c) `=NB.SI(B2:B10 ; "Bouaké")`

👉 Voir la réponse

✓ b) `SOMME.SI()` / `SUMIF()` — Elle additionne la plage D2:D10 uniquement pour les lignes où B2:B10 = "Bouaké". `SOMME()` additionnerait tout sans condition. `NB.SI()` compte, ne somme pas.

Q2. Ta formule `=RECHERCHEV(A2 ; G2:H10 ; 2 ; 0)` retourne `#N/A` pour certaines lignes. Comment corriger ?

- a) Changer le 0 en 1
- b) Envelopper dans `=SIERREUR(RECHERCHEV(A2;G2:H10;2;0) ; "Non trouvé")`
- c) Supprimer les lignes avec #N/A

👉 Voir la réponse

✓ b) `SIERREUR()` / `IFERROR()` — `#N/A` signifie que la valeur n'a pas été trouvée dans le tableau. `SIERREUR` remplace l'erreur par un message propre. Changer 0 en 1 activerait la correspondance approximative — dangereux et incorrect ici.

Q3. Tu copies la formule `=B2*E1` vers le bas. E1 contient le taux de commission fixe. Après la copie, E1 est devenu E2, E3... Le résultat est faux. Quelle est la correction ?

- a) `=B2*$E$1` ou `=B2*$E$1`
- b) `=B2*E1` — la formule est correcte
- c) Recopier la formule manuellement

👉 Voir la réponse

✓ a) `=$B2*$E$1` — Il faut figer E1 avec `$` pour qu'il ne bouge pas à la copie. `$E$1` fige colonne ET ligne. `E$1` fige seulement la ligne — suffisant ici car on copie vers le bas. Raccourci : sélectionner E1 dans la barre de formule et appuyer sur F4.

Q4. Tu veux afficher "🟢 Atteint" si les ventes $\geq$ objectif, "🔴 Manqué" sinon. Quelle formule ?

- a) `=SOMME.SI(D2>=E2 ; "🟢 Atteint" ; "🔴 Manqué")`
- b) `=SI(D2>=E2 ; "🟢 Atteint" ; "🔴 Manqué")`
- c) `=NB.SI(D2 ; ">="&E2)`

👉 Voir la réponse

✅ b) `SI()` / `IF()` — C'est exactement l'usage de SI : si condition vraie $\rightarrow$ valeur 1, sinon $\rightarrow$ valeur 2. `SOMME.SI` additionne selon condition, elle ne retourne pas de texte.

Q5. La colonne Région contient "Abidjan " (avec un espace invisible à la fin). Ta formule

`=NB.SI(B:B ; "Abidjan")` retourne 0. Quelle est la cause et la solution ?

- a) La formule est incorrecte
- b) Les espaces invisibles font que "Abidjan " $\neq$ "Abidjan" — utiliser `SUPPRESSESPACE()` / `TRIM()` pour nettoyer
- c) NB.SI ne fonctionne pas avec du texte

👉 Voir la réponse

✅ b) `SUPPRESSESPACE()` / `TRIM()` — Les espaces invisibles sont l'un des pièges les plus fréquents en analyse de données. "Abidjan " (avec espace) est différent de "Abidjan" pour Excel. Appliquer `=SUPPRESSESPACE(B2)` / `=TRIM(B2)` sur toute la colonne avant l'analyse.

Q6. Tu veux créer une clé unique combinant le nom du vendeur et sa région pour rapprocher deux tableaux. Quelle formule ?

- a) `=SOMME(A2 ; B2)`
- b) `=A2 & "_" & B2`
- c) `=SI(A2=B2 ; A2 ; B2)`

Voir la réponse

✓ b) `=A2 & "_" & B2` — L'opérateur `&` concatène du texte. Le résultat sera `"Adjoua_Abidjan"` — une clé unique utilisable dans un `RECHERCHEV()` / `VLOOKUP()` pour rapprocher deux tableaux qui n'ont pas de clé commune.

Module suivant

Tu maîtrises maintenant les formules essentielles. Dans le **Module 05**, on passe aux Tableaux Croisés Dynamiques — l'outil qui permet de résumer et analyser n'importe quel dataset en quelques clics.

→ [Module 05 — TCD & Visualisation Excel](#)